



## **Precession in the General Theory of Relativity**

### **Synopsis**

This is a relatively quick and informal sketch of a demonstration that general relativistic corrections to the bound Kepler orbits introduce a perihelion precession. Any decent textbook on the general theory of relativity will derive this result. My analysis aligns with that found in the good old text "Introduction to General Relativity", by Adler, Bazin and Schiffer.

The plan of the analysis is as follows.

- \* Model the planetary orbits as geodesics in the (exterior) Schwarzschild spacetime.
- \* Compute the geodesic equations.
- \* Simplify them using symmetries and first integrals.
- \* Isolate the differential equation expressing the radial coordinate as a function of orbital angle - the "equation of the orbit".
- \* Use a perturbative approach to isolate the perihelion precession.
- \* The perturbative solution of the equation of the orbit involves one term which is not periodic in the orbital angle - this (small) term is responsible for the perihelion precession.

## Define the Schwarzschild Spacetime

I will use the DifferentialGeometry package and its Tensor subpackage.

```
> with(DifferentialGeometry): with(Tensor):  
> interface(typesetting = extended):  
> Preferences("TensorDisplay", 1):
```

Begin by defining the Schwarzschild spacetime with mass  $m$ . This command creates a coordinate chart (which will be the usual Schwarzschild coordinate chart) and names it "M".

```
> DGsetup([t, r, theta, phi], M);
```

*Manifold: M* (1)

This command creates the metric as a rank-2 tensor. (Notice that we are using "geometric units".)

```
M > g := evalDG(-(1- 2*m/r)*dt &t dt + (1- 2*m/r)^(-1)*dr &t dr + r^2*(dtheta &t dtheta +  
sin(theta)^2*dphi &t dphi));
```

$$g := \frac{-r + 2m}{r} dt \otimes dt - \frac{r}{-r + 2m} dr \otimes dr + r^2 d\theta \otimes d\theta + r^2 \sin(\theta)^2 d\phi \otimes d\phi$$

(2)

This metric solves the vacuum Einstein equations,

```
M > EinsteinTensor(g);
```

$0 \partial_t \otimes \partial_t$  (3)

It has Petrov type D,

```
M > PetrovType(g);
```

"D" (4)

It has a 4 dimensional symmetry group consisting of time translations and spatial rotations (for  $r > 2m$ ).

**M > KV := KillingVectors(g);**

$$KV := \left[ \frac{\sqrt{-\cos(2\theta) + 1}}{\sin(\theta)} \partial_{\phi'} - \frac{\sqrt{-\cos(2\theta) + 1} \cos(\phi)}{\sin(\theta)} \partial_{\theta} + \frac{\sqrt{-\cos(2\theta) + 1} \cos(\theta) \sin(\phi)}{\sin(\theta)^2} \partial_{\phi'} \right. \\ \left. \frac{\sqrt{-\cos(2\theta) + 1} \sin(\phi)}{\sin(\theta)} \partial_{\theta} + \frac{\sqrt{-\cos(2\theta) + 1} \cos(\theta) \cos(\phi)}{\sin(\theta)^2} \partial_{\phi'} - \partial_t \right] \quad (5)$$

## Compute the Geodesic Equations

Construct the geodesic equation for a curve parameterized with the proper time  $s$ . The acceleration of the curves, denoted by the vector  $A$ , must vanish if the curves are geodesics. The proper time parametrization condition will be dealt with in the next section.

**M > C := Christoffel(g);**

**M > curve := [t(s), r(s), theta(s), phi(s)];**

$$curve := [t(s), r(s), \theta(s), \phi(s)] \quad (6)$$

**M > A := GeodesicEquations(curve, C, s);**

$$A := \left( \frac{d^2}{ds^2} t(s) + \frac{2 \left( \frac{d}{ds} t(s) \right) \left( \frac{d}{ds} r(s) \right) m}{r(s) (r(s) - 2m)} \right) \partial_t + \frac{1}{r(s)^3 (-r(s) + 2m)} \left( (2m r(s)^3 - r(s)^4) \left( \frac{d^2}{ds^2} r(s) \right) - 4 \left( -\frac{r(s)}{2} + m \right)^2 r(s)^3 (\cos(\theta(s)) + 1) (\cos(\theta(s)) - 1) \left( \frac{d}{ds} \phi(s) \right)^2 + 4 \left( -\frac{r(s)}{2} + m \right)^2 r(s)^3 \left( \frac{d}{ds} \theta(s) \right)^2 - 4 \left( -\frac{r(s)}{2} + m \right)^2 m \left( \frac{d}{ds} t(s) \right)^2 + \left( \frac{d}{ds} r(s) \right)^2 m r(s)^2 \right) \partial_r \\ + \frac{- \left( \frac{d}{ds} \phi(s) \right)^2 \cos(\theta(s)) \sin(\theta(s)) r(s) + 2 \left( \frac{d}{ds} r(s) \right) \left( \frac{d}{ds} \theta(s) \right) + \left( \frac{d^2}{ds^2} \theta(s) \right) r(s)}{r(s)} \partial_{\theta} \quad (7)$$

$$+ \frac{2 \left( \frac{d}{ds} \theta(s) \right) \left( \frac{d}{ds} \phi(s) \right) \cos(\theta(s)) r(s) + 2 \left( \frac{d}{ds} r(s) \right) \left( \frac{d}{ds} \phi(s) \right) \sin(\theta(s)) + \left( \frac{d^2}{ds^2} \phi(s) \right) r(s) \sin(\theta(s))}{r(s) \sin(\theta(s))} \partial_\phi$$

Extract the components of the acceleration. Setting these to zero constructs the system of ODEs which are the geodesic equations and are denoted by  $Geq = 0$ .

**M > Geq := GetComponents(A, [D\_t, D\_r, D\_theta, D\_phi]);**

$$Geq := \left[ \frac{2 \left( \frac{d}{ds} t(s) \right) \left( \frac{d}{ds} r(s) \right) m + \left( \frac{d^2}{ds^2} t(s) \right) r(s)^2 - 2 \left( \frac{d^2}{ds^2} t(s) \right) r(s) m}{r(s) (r(s) - 2m)}, -\frac{1}{r(s)^3 (r(s) - 2m)} \left( -r(s)^5 \left( \frac{d}{ds} \phi(s) \right)^2 \cos(\theta(s))^2 + 4 r(s)^4 \left( \frac{d}{ds} \phi(s) \right)^2 \cos(\theta(s))^2 m - 4 r(s)^3 \left( \frac{d}{ds} \phi(s) \right)^2 \cos(\theta(s))^2 m^2 + r(s)^5 \left( \frac{d}{ds} \phi(s) \right)^2 \right. \right. \quad (8)$$

$$+ r(s)^5 \left( \frac{d}{ds} \theta(s) \right)^2 - 4 r(s)^4 \left( \frac{d}{ds} \phi(s) \right)^2 m - 4 r(s)^4 \left( \frac{d}{ds} \theta(s) \right)^2 m + 4 r(s)^3 \left( \frac{d}{ds} \phi(s) \right)^2 m^2 + 4 r(s)^3 \left( \frac{d}{ds} \theta(s) \right)^2 m^2 + \left. \left( \frac{d}{ds} r(s) \right)^2 m r(s)^2 - \left( \frac{d}{ds} t(s) \right)^2 r(s)^2 m + 4 \left( \frac{d}{ds} t(s) \right)^2 r(s) m^2 - 4 \left( \frac{d}{ds} t(s) \right)^2 m^3 - \left( \frac{d^2}{ds^2} r(s) \right) r(s)^4 + 2 \left( \frac{d^2}{ds^2} r(s) \right) r(s)^3 m \right),$$

$$\left. - \frac{\left( \frac{d}{ds} \phi(s) \right)^2 \cos(\theta(s)) \sin(\theta(s)) r(s) + 2 \left( \frac{d}{ds} r(s) \right) \left( \frac{d}{ds} \theta(s) \right) + \left( \frac{d^2}{ds^2} \theta(s) \right) r(s)}{r(s)}, \right.$$

$$\left[ \frac{2 \left( \frac{d}{ds} \theta(s) \right) \left( \frac{d}{ds} \phi(s) \right) \cos(\theta(s)) r(s) + 2 \left( \frac{d}{ds} r(s) \right) \left( \frac{d}{ds} \phi(s) \right) \sin(\theta(s)) + \left( \frac{d^2}{ds^2} \phi(s) \right) r(s) \sin(\theta(s))}{r(s) \sin(\theta(s))} \right]$$

## Simplify the Equations Via Symmetries and First Integrals

It is convenient to replace the second geodesic equation,  $Geq[2]$ , which involves the second derivative of  $r(s)$ , with the proper time parametrization equation,  $g(W, W) = -1$ , where  $W$  is the tangent vector to the geodesic - the 4-velocity of the planet. This parametrization/normalization condition is a first integral of the geodesic equation and it only involves the first derivative of  $r(s)$ .

Here is the tangent to the curve:

```
M > W := evalDG(diff(t(s),s)*D_t + diff(r(s),s)*D_r + diff(theta(s),s)*D_theta + diff(phi(s),s)*D_phi);
```

$$W := \frac{d}{ds} t(s) \partial_t + \frac{d}{ds} r(s) \partial_r + \frac{d}{ds} \theta(s) \partial_\theta + \frac{d}{ds} \phi(s) \partial_\phi \quad (9)$$

Here we write down the left hand side of the parametrization condition  $g(W, W) + 1 = 0$ , evaluated on the curve. We denote it by  $vel\_eq$ .

```
M > vel_eq := simplify(Hook([W, W], eval(g, [r=r(s), theta=theta(s)])) + 1);
```

$$vel\_eq := \frac{1}{r(s) (-r(s) + 2m)} \left( -2 r(s)^3 (\cos(\theta(s)) + 1) (\cos(\theta(s)) - 1) \left( -\frac{r(s)}{2} + m \right) \left( \frac{d}{ds} \phi(s) \right)^2 + 4 \left( -\frac{r(s)}{2} + m \right)^2 \left( \frac{d}{ds} t(s) \right)^2 + 2 \left( r(s)^2 \left( -\frac{r(s)}{2} + m \right) \left( \frac{d}{ds} \theta(s) \right)^2 - \frac{\left( \frac{d}{ds} r(s) \right)^2 r(s)}{2} + m - \frac{r(s)}{2} \right) r(s) \right) \quad (10)$$

Here, then, are the left-hand sides of the simplified geodesic conditions, denoted by  $Geq1=0$ :

```
M > Geq1 := [Geq[1], vel_eq, Geq[3], Geq[4]];
```

$$\begin{aligned}
\text{Geq1} := & \left[ \frac{2 \left( \frac{d}{ds} t(s) \right) \left( \frac{d}{ds} r(s) \right) m + \left( \frac{d^2}{ds^2} t(s) \right) r(s)^2 - 2 \left( \frac{d^2}{ds^2} t(s) \right) r(s) m}{r(s) (r(s) - 2m)}, \frac{1}{r(s) (-r(s) + 2m)} \left( \right. \right. \\
& -2 r(s)^3 (\cos(\theta(s)) + 1) (\cos(\theta(s)) - 1) \left( -\frac{r(s)}{2} + m \right) \left( \frac{d}{ds} \phi(s) \right)^2 + 4 \left( -\frac{r(s)}{2} + m \right)^2 \left( \frac{d}{ds} t(s) \right)^2 \\
& + 2 \left( r(s)^2 \left( -\frac{r(s)}{2} + m \right) \left( \frac{d}{ds} \theta(s) \right)^2 - \frac{\left( \frac{d}{ds} r(s) \right)^2 r(s)}{2} + m - \frac{r(s)}{2} \right) r(s) \left. \right), \\
& \frac{-\left( \frac{d}{ds} \phi(s) \right)^2 \cos(\theta(s)) \sin(\theta(s)) r(s) + 2 \left( \frac{d}{ds} r(s) \right) \left( \frac{d}{ds} \theta(s) \right) + \left( \frac{d^2}{ds^2} \theta(s) \right) r(s)}{r(s)}, \\
& \left. \frac{2 \left( \frac{d}{ds} \theta(s) \right) \left( \frac{d}{ds} \phi(s) \right) \cos(\theta(s)) r(s) + 2 \left( \frac{d}{ds} r(s) \right) \left( \frac{d}{ds} \phi(s) \right) \sin(\theta(s)) + \left( \frac{d^2}{ds^2} \phi(s) \right) r(s) \sin(\theta(s))}{r(s) \sin(\theta(s))} \right] \quad (11)
\end{aligned}$$

The time translation and rotational symmetries of the spacetime metric imply corresponding symmetries and conservation laws for the geodesic equations. We now simplify the geodesic equations using conservation of energy, and by using rotational symmetry and conservation of angular momentum.

To begin, conservation of (the direction of) angular momentum keeps the motion in a (spacelike) plane, just as in the non-relativistic, Newtonian analysis. Taking account of rotational symmetry, we lose no generality if we choose the plane of the orbit to be the plane given by  $\theta = \frac{\pi}{2}$ . The resulting equations are denoted  $\text{Geq2} = 0$ . Using this specialization, the third ( $\theta$ )

equation of  $\text{Geq1}$  is satisfied, and the fourth ( $\phi$ ) equation simply says the magnitude of "angular momentum",  $L = r^2 \frac{d\phi}{ds}$ , is conserved.

**M > Geq2 := eval(Geq1, theta(s) = Pi/2);**

$$\begin{aligned}
 \text{Geq2} := & \left[ \frac{2 \left( \frac{d}{ds} t(s) \right) \left( \frac{d}{ds} r(s) \right) m + \left( \frac{d^2}{ds^2} t(s) \right) r(s)^2 - 2 \left( \frac{d^2}{ds^2} t(s) \right) r(s) m}{r(s) (r(s) - 2m)}, \right. \\
 & \frac{2 r(s)^3 \left( -\frac{r(s)}{2} + m \right) \left( \frac{d}{ds} \phi(s) \right)^2 + 4 \left( -\frac{r(s)}{2} + m \right)^2 \left( \frac{d}{ds} t(s) \right)^2 + 2 \left( -\frac{\left( \frac{d}{ds} r(s) \right)^2 r(s)}{2} + m - \frac{r(s)}{2} \right) r(s)}{r(s) (-r(s) + 2m)}, \\
 & \left. 0, \frac{2 \left( \frac{d}{ds} r(s) \right) \left( \frac{d}{ds} \phi(s) \right) + \left( \frac{d^2}{ds^2} \phi(s) \right) r(s)}{r(s)} \right]
 \end{aligned} \tag{12}$$

**M > dsolve(Geq2[4], phi(s));**

$$\phi(s) = \_C1 + \left( \int \frac{1}{r(s)^2} ds \right) \_C2 \tag{13}$$

We can now eliminate  $\phi(s)$  via

$$\frac{d\phi}{ds} = \frac{L}{r^2},$$

to get a reduced system of equations for  $t(s)$  and  $r(s)$ , which we denote by  $\text{Geq3} = 0$ .

**M > Geq3 := eval(Geq2, {diff(phi(s), s)= L/r(s)^2, diff(phi(s), s, s)= diff(L/r(s)^2, s)});**

$$\text{Geq3} := \left[ \frac{2 \left( \frac{d}{ds} t(s) \right) \left( \frac{d}{ds} r(s) \right) m + \left( \frac{d^2}{ds^2} t(s) \right) r(s)^2 - 2 \left( \frac{d^2}{ds^2} t(s) \right) r(s) m}{r(s) (r(s) - 2m)}, \right.
 \end{aligned} \tag{14}$$

$$\left[ \frac{2 \left( -\frac{r(s)}{2} + m \right) L^2}{r(s)} + 4 \left( -\frac{r(s)}{2} + m \right)^2 \left( \frac{d}{ds} t(s) \right)^2 + 2 \left( -\frac{\left( \frac{d}{ds} r(s) \right)^2 r(s)}{2} + m - \frac{r(s)}{2} \right) r(s)}{r(s) (-r(s) + 2m)}, 0, 0 \right]$$

## Obtain the Equation of the Orbit

The time translation symmetry allows us to integrate the equation for  $t(s)$ , which is  $Geq3[1] = 0$ , and eliminate  $t(s)$  in favor of  $r(s)$ .

**M > tsol := dsolve(Geq3[1], t(s));**

$$tsol := t(s) = \_C1 + \left( \int \frac{r(s)}{r(s) - 2m} ds \right) \_C2 \quad (15)$$

We substitute this solution into  $Geq3$  to obtain the final differential equation for  $r(s)$ , which (with a little cleaning) we denote by  $Geq4 = 0$ .

**M > simplify(eval(Geq3, tsol));**

$$\left[ 0, \frac{\left( \frac{d}{ds} r(s) \right)^2 r(s)^3 - r(s)^3 \_C2^2 + L^2 r(s) - 2 L^2 m + r(s)^3 - 2 m r(s)^2}{(r(s) - 2m) r(s)^2}, 0, 0 \right] \quad (16)$$

**M > Geq4 := numer(simplify(eval(Geq3[2], tsol)));**

$$Geq4 := \left( \frac{d}{ds} r(s) \right)^2 r(s)^3 - r(s)^3 \_C2^2 + L^2 r(s) - 2 L^2 m + r(s)^3 - 2 m r(s)^2 \quad (17)$$

As usual in orbital mechanics, the shape of the orbit is easier to characterize than its explicit time evolution. We therefore trade  $s$  for  $\phi$  via



$$\frac{dr}{ds} = \frac{dr}{d\phi} \frac{d\phi}{ds} = \frac{dr}{d\phi} \frac{L}{r^2}$$

We are thus computing the path of the orbit  $r = r(\phi)$  instead of its proper time evolution ( $r = r(s)$ ,  $\phi(s)$ ). After some cleaning up by algebraic manipulations, the equation for the path is denoted by  $req3 = 0$ .

**M > req2 := expand(Geq4/r(s)^3);**

$$req2 := \left( \frac{d}{ds} r(s) \right)^2 - C^2 + \frac{L^2}{r(s)^2} - \frac{2 L^2 m}{r(s)^3} + 1 - \frac{2 m}{r(s)} \quad (18)$$

**M > req3:=expand(r(phi)^4/L^2\*simplify(eval(eval(req2, diff(r(s),s) = diff(r(phi),phi)\*L/r(s)^2), r(s) = r(phi))));**

$$req3 := \left( \frac{d}{d\phi} r(\phi) \right)^2 - \frac{r(\phi)^4 - C^2}{L^2} + r(\phi)^2 - 2 r(\phi) m + \frac{r(\phi)^4}{L^2} - \frac{2 r(\phi)^3 m}{L^2} \quad (19)$$

Finally, as in the Newtonian 2-body problem, it is simpler to work with  $u = 1/r$ . After changing variables (and some more simplifying) we finally get the orbit equation for  $u$ , denoted by  $Ueq = 0$ .

**M > Ueq := expand(u(phi)^4\*simplify(eval(req3, r(phi) = 1/u(phi))));**

$$Ueq := -2 m u(\phi)^3 + u(\phi)^2 + \left( \frac{d}{d\phi} u(\phi) \right)^2 - \frac{2 m u(\phi)}{L^2} - \frac{C^2}{L^2} + \frac{1}{L^2} \quad (20)$$

To facilitate comparison with the familiar Newtonian result, we differentiate  $Ueq$  and use the resulting second-order equation, denoted by  $Ueq1$ .

**M > Ueq1 := expand(1/2/diff(u(phi), phi)\*diff(Ueq, phi));**

$$Ueq1 := -3 m u(\phi)^2 + u(\phi) + \frac{d^2}{d\phi^2} u(\phi) - \frac{m}{L^2} \quad (21)$$

It is possible to solve this equation, but the solution is in terms of an integral and it is not immediately obvious what it all means. We will use an approximation method to isolate the behavior we are interested in.

**M > dsolve(Ueq1);**

$$\int_{-\phi - C2}^{u(\phi)} \frac{L}{\sqrt{2 m a^3 L^2 - L^2 a^2 + C1 L^2 + 2 a m}} d_a - \phi - C2 = 0, \int_{-\phi - C2}^{u(\phi)} - \frac{L}{\sqrt{2 m a^3 L^2 - L^2 a^2 + C1 L^2 + 2 a m}} d_a \quad (22)$$

## Use Perturbation Theory to Isolate the Non-Newtonian Part

The orbit equation for  $u(\phi)$  can be solved by quadrature, but this is not immediately enlightening. To see the perihelion precession, we note that the term  $-3 m u^2$  distinguishes the orbit equation from its exactly soluble Newtonian version. The  $-3 m u^2$  term is in fact very small compared to  $\frac{m}{L^2}$ , provided the speed of the planet is non-relativistic, so we isolate its effects perturbatively using  $\epsilon$  as a dimensionless parameter, which we set to unity at the end of the calculation. The orbit equation in terms of  $\epsilon$  is denoted by  $Ueq2 = 0$ .

**M > Ueq2 := eval(Ueq1, [m = epsilon\*m, L = sqrt(epsilon)\*L]);**

$$Ueq2 := -3 \epsilon m u(\phi)^2 + u(\phi) + \frac{d^2}{d\phi^2} u(\phi) - \frac{m}{L^2} \quad (23)$$

We write  $u = u0 + \epsilon v$ , where  $u0$  will satisfy the Newtonian form of the orbit equation (the  $\epsilon = 0$  part) and  $v$  will represent the relativistic correction.

**M > Ueq3 := eval(Ueq2, u(phi) = u0(phi) + epsilon\*v(phi));**

$$Ueq3 := -3 \epsilon m (u0(\phi) + \epsilon v(\phi))^2 + u0(\phi) + \epsilon v(\phi) + \frac{d^2}{d\phi^2} u0(\phi) + \epsilon \left( \frac{d^2}{d\phi^2} v(\phi) \right) - \frac{m}{L^2} \quad (24)$$

We solve this equation to first order in epsilon. Here is the zeroth order equation for  $u0$ , which is the Newtonian orbit equation

for the inverse square force law.

**M > u0eq := eval(Ueq3, epsilon=0);**

$$u0eq := u0(\phi) + \frac{d^2}{d\phi^2} u0(\phi) - \frac{m}{L^2} \quad (25)$$

Here is the first order equation for v.

**M > veq := eval(diff(Ueq3, epsilon), epsilon=0);**

$$veq := -3 m u0(\phi)^2 + v(\phi) + \frac{d^2}{d\phi^2} v(\phi) \quad (26)$$

We solve these equations. This is the zeroth-order Newtonian solution. Note, in particular, that this solution has  $u$  being a periodic function of the angle, which implies the zeroth order orbits are closed.

**M > u0\_solution:=eval(dsolve(u0eq ), [\_C1 = B, \_C2=0]);**

$$u0\_solution := u0(\phi) = \cos(\phi) B + \frac{m}{L^2} \quad (27)$$

This is the (approximate) relativistic correction:

**M > veq1 := expand(simplify(eval(veq, u0\_solution)));**

$$veq1 := -3 B^2 \cos(\phi)^2 m - \frac{6 B \cos(\phi) m^2}{L^2} + v(\phi) + \frac{d^2}{d\phi^2} v(\phi) - \frac{3 m^3}{L^4} \quad (28)$$

**M > v\_solution:=dsolve(veq1);**

$$v\_solution := v(\phi) = \sin(\phi) \_C2 + \cos(\phi) \_C1 - \frac{m (B^2 L^4 \cos(\phi)^2 - 2 B^2 L^4 - 3 B L^2 \sin(\phi) m \phi - 3 B L^2 \cos(\phi) m - 3 m^2)}{L^4} \quad (29)$$

**M >**

If  $u$  is a periodic function of the angle  $\phi$ , then so is  $r$  and the orbits are closed. The perihelion precession occurs because  $u$  is not a periodic function of  $\phi$ , which occurs due to the term involving  $\frac{3 Bm^2}{L^2} \phi \sin(\phi)$  in  $v$  in this approximation. In this approximation, when  $\phi$  advances by  $2\pi$ , the perihelion advances by  $\frac{6\pi Bm^2}{L^2}$ . This is what yields the famous perihelion shift of 43.03 seconds of arc per century for the planet Mercury.

#### Commands Illustrated

[EinsteinTensor](#), [PetrovType](#), [KillingVectors](#), [Christoffel](#), [GeodesicEquations](#), [GetComponents](#)

#### References

"Introduction to General Relativity", by Adler, Bazin and Schiffer, McGraw-Hill, Second Edition, 1975

#### Release Notes

Worksheet was executed using Maple 2018 and DifferentialGeometry build "USU001-1460 9:25:48.66 Fri 01/03/2020"

#### Author

Charles Torre  
Utah State University  
4/4/17